

## Введение

В 1980 году группа исследователей канадского университета Waterloo занялась проблемой создания компьютерной системы, эффективной в решении алгебраических задач и достаточно простой для того, чтобы её могли использовать не только математики и инженеры, но и студенты. Программа получила имя **Maple**. В начале 90-х у Maple появился графический интерфейс пользователя, и именно с этого времени Maple стал широко применяться в образовании.

Waterloo Maple, наряду с Wolfram Mathematica, является мощнейшей системой символьной математики (или компьютерной алгебры). Даже в тех случаях, когда вычисления носят символьный характер, расчётные алгоритмы реализуются так, чтобы получить сначала аналитический результат. В Maple в общей сложности используется более трёх тысяч команд, однако некоторые из них (относящиеся к проблемам интегрирования, дифференцирования функций, решения уравнений и т.п.) применяются достаточно часто и составляют костяк базового языка. Некоторые команды доступны только при подключении специальных пакетов.

Данное пособие является практическим руководством и ориентировано на описание командного языка Maple. Работа с пособием предполагает знание, хотя бы в минимальном объёме, основ высшей математики. В пособии приведено много разнообразных примеров решения задач линейной алгебры, математического анализа, дифференциальных уравнений, математической физики и численных методов.

В Maple предусмотрена мощная и эффективная система справки. Справку по любой команде можно получить, наведя курсор на эту команду и затем нажав <F1>. Ещё один способ – в области ввода ввести знак вопроса и, без пробела, название интересующей пользователя команды, после чего нажать <Enter>.

## 1. Знакомство с Maple

При запуске Maple автоматически создаётся новый документ, который в дальнейшем будем называть рабочим листом. В области ввода пользователь видит символ начала командной строки (>) – здесь осуществляется ввод команд. Результат выполнения команд отображается в том же рабочем листе, в области вывода. Если команда заканчивается точкой с запятой (;), то результат выполнения появится в ячейке вывода, если же в конце команды поставить двоеточие (:), результат выводиться не будет.

Ввод фиксируется нажатием клавиши <Enter>. Если надо перенести ввод на новую строку, используется комбинация клавиш <Shift> + <Enter>.

В системной переменной % хранится результат предыдущей операции. С помощью одного, двух или трёх знаков % можно вызвать первое, второе или третье выражение с конца сессии.

Для присваивания переменной значения используется оператор :=. Maple сохраняет в памяти все определения и присваивания, которые были сделаны во всех загруженных в систему документах. Для очистки внутренней памяти Maple используют команду restart. В примерах пособия эта команда не используется, однако нужно иметь в виду, что решение каждой новой задачи следует начинать с команды restart.

### Лабораторная работа №1.

#### Знакомство с Maple.

Для вычисления определённого интеграла используется функция int:

```
> int(2 * exp(-x) * sin(x), x = 0 .. infinity);
1
```

Рассмотрим последовательность команд:

```
> x := 2 * exp(-1) * sin(1);
x := e(-1) sin(1)
> int(2 * exp(-x) * sin(x), x = 0 .. infinity);
Error, (in int) wrong number (or type) of arguments
```

Где допущена ошибка? Как её исправить?

### **Инертные формы функций**

Сравните две следующие команды:

```
> int(2 * exp(-x) * sin(x), x = 0 .. infinity);
1
> Int(2 * exp(-x) * sin(x), x = 0 .. infinity);
∫0∞ 2e(-x) sin(x) dx
```

Int – это так называемая инертная форма функции int. Имена инертных функций начинаются с большой буквы, эти функции выводят выражение в естественном математическом виде. С помощью ряда функций, например evalf, можно вычислить выражение, полученное инертной функцией:

```
> Limit(sin(x) / x, x = 0);
limx→0  $\frac{\sin(x)}{x}$ 
> evalf(%);
1.000000000
```

Иногда используют следующую запись:

```
> Limit((1 + 1 / n) ^ n, n = infinity) =
limit((1 + 1 / n) ^ n, n = infinity);
limn→∞  $\left(1 + \frac{1}{n}\right)^n = e$ 
```

## 2. Типы данных, операторы и функции системы Maple

### Выражения и основы работы над ними

Пользовательский интерфейс системы Maple позволяет представлять как вводимые, так и выводимые выражения в самых различных формах, в том числе и в естественном математическом виде. В данном пособии выражения записываются на Maple-языке без использования специальных средств для их представления. При наборе математических выражений удобно пользоваться палитрами математических символов (меню **View à Palettes**).

Для упрощения выражений используют функцию `simplify`, а для их расширения – функцию `expand`:

```
> simplify(sin(x)^2 + cos(x)^2);
1
> expand(sin(2 * x));
2 sin(x)cos(x)
```

Для разложения выражения на множители используют функцию `factor`, а для комплектования по степеням – функцию `collect`:

```
> factor(x^2 + 2 * x * y + y^2);
(x + y)^2
> factor(x^2 + 2 * x + 2, complex);
(x + 1.000000000 + 1.000000000 I)(x + 1. - 1. I)
> g := int(x * (exp(x) - exp(-x)), x);
g := x e^x - e^x + x/e^x + 1/e^x
> collect(g, exp(x));
g := (x - 1)e^x + (x + 1)/e^x
```

Для преобразования выражений в тождественные формы используют функцию `convert`:

```
> convert(2 * sin(I * x) + 2 * sinh(x), exp);
2 I (1/2 e^x - 1/2 1/e^x) + e^x + e^x - 1/e^x
> collect(% , exp);
(I + I)e^x - (I + I)/e^x
> convert(arcsinh(x), ln);
ln(x + sqrt(x^2 + 1))
```

Некоторые параметры функции `convert` представлены в **Приложении 1**.

Для создания последовательностей выражений используют функцию `seq`:

```
> seq(sin(x), x = 0 .. 5);
0, sin(1), sin(2), sin(3), sin(4), sin(5)
```

Для оценивания выражений различного типа существует группа функций, основные из которых представлены в **Приложении 1**.

```
> eval(x^2 - x * y + y^2, [x = 1/3, y = 1/6]);
1
12
> evalf(pi, 20);
3.1415926535897932385
> evalr(exp(INTERVAL(0 .. ln(2)))));
INTERVAL(1..2)
```

### Лабораторная работа №2.

#### Оценивание выражений.

С помощью функции `shake` оценим интервал, в пределах которого находится число  $p$ :

```
> shake(Pi, 5);
INTERVAL(3.141279 .. 3.141907)
```

Получите такой же результат, используя функцию `evalr`.

Обоими способами оцените интервалы, в пределах которых находятся следующие выражения:

```
arctg(2);
e(-1) sin(1);
cosh(p) - sinh(p).
```

### **Контроль за типами объектов**

Функция `whattype` возвращает тип объекта:

```
> whattype(Pi);
symbol
> whattype(2 + 3);
integer
> whattype(sin(x));
function
```

С помощью функции `type` можно выяснить, относится ли указанный объект к соответствующему типу:

```
> type(Pi, constant);
true
> type(sin(ln(x)), function);
true
```

Основные типы данных языка Maple перечислены в **Приложении 2**.

## Простые типы данных

Maple работает с числами следующего типа: целыми, рациональными, вещественными с плавающей точкой и комплексными (подробный список числовых типов данных можно найти в **Приложении 2**). Указание десятичной точки в числе делает его вещественным и ведёт к переводу вычислений в режим работы с вещественными числами. При этом количеством выводимых после десятичной точки цифр можно управлять, задавая значение системной переменной `Digits`:

```
> exp(-1) * sin(1);

$$e^{(-1)} \sin(1)$$

> Digits := 10:
  exp(-1.) * sin(1.);
0.3095598757
```

Мнимая единица обозначается в Maple как `I`.

Для контроля за числами используют уже известные функции `whattype` и `type`:

```
> type(30, numeric);
true
> type(30, integer);
true
> type(30.0, integer);
false
> whattype(30.0);
float
> type(Pi * I, complex);
true
```

Функция `convert` служит для преобразования чисел с разными основаниями, а также для перевода чисел в разные форматы (рациональная дробь или вещественное число с плавающей точкой):

```
> convert(30, binary);
11110
> convert(11110, decimal, binary);
30
> convert(1 / 5, float);
0.2000000000
> convert(0.2, rational);

$$\frac{1}{5}$$

> convert(Pi, float);
3.141592654
```

Некоторые параметры функции `convert` описаны в **Приложении 1**.

Для преобразования единиц измерения используют команду **Unit Converter** (меню **Edit à Unit Converter**).

Основные математические функции представлены в **Приложении 3**.

## Данные множественного типа

Неупорядоченные наборы – множества – создаются с помощью фигурных скобок. Из множеств автоматически удаляются повторяющиеся элементы.

```
> {a, b, g, a, a, c, c, f};
{f,g,a,b,c}
> {5, 7 - 4, pi, exp(ln(3)), sqrt(25), 3!};
{3,5,6,p}
```

Операторы для работы с данными множественного типа представлены в **Приложении 6**.

```
> {a, b, g, a, a, c, c, f} intersect {a, b, c, d};
{a,b,c}
```

Упорядоченные наборы – списки – создаются с помощью квадратных скобок:

```
> s1 := [5, 7 - 4, pi, exp(ln(3)), sqrt(25), 3!];
s1:= [5,3,p,3,5,6]
> s2 := [[a, b], [c, d]];
s3 := [[1, 2], [2, 5]];
s2:= [[a,b],[c,d]]
s3:= [[1,2],[2,5]]
```

Списки широко применяются для задания векторов и матриц:

```
> V := array(1 .. 6, s1);
V:= [5,3,p,3,5,6]
> M1 := array(1 .. 2, 1 .. 2, s2);
M2 := convert(s3, matrix);
M1:=  $\begin{bmatrix} a & b \\ c & d \end{bmatrix}$ 
M2:=  $\begin{bmatrix} 1 & 2 \\ 2 & 5 \end{bmatrix}$ 
```

Элементы векторов и матриц являются индексированными переменными. Допустимы операции вызова нужного элемента и присваивания ему нового значения.

```
> V[1] := V[3] * V[6] / 3;
evalm(V);
[2p,3,p,3,5,6]
> evalm(2 * V);
[4p,6,2p,6,10,12]
> evalm(M1 - a * M2);
 $\begin{bmatrix} 0 & b-2a \\ c-2a & d-5a \end{bmatrix}$ 
```

```
> evalm(M1 &* M2);

$$\begin{bmatrix} a+2b & 2a+5b \\ c+2d & 2c+5d \end{bmatrix}$$

> evalm(M2^(-1));

$$\begin{bmatrix} 5 & -2 \\ -2 & 1 \end{bmatrix}$$

```

Функция `map` применяет заданную операцию к каждому элементу матрицы или вектора:

```
> M := convert([[x, x^2], [0, 1 + x]], matrix);
  map(sin, M);

$$\begin{bmatrix} \sin(x) & \sin(x^2) \\ 0 & \sin(1+x) \end{bmatrix}$$

> map(diff, M, x);

$$\begin{bmatrix} 1 & 2x \\ 0 & 1 \end{bmatrix}$$

```

В ядро Maple введены минимально необходимые средства для работы с векторами и матрицами. Основной упор сделан на пакеты расширения, главным из которых является пакет `linalg`. Некоторые функции из этого пакета представлены в **Приложении 4**.

Для загрузки пакета используют команду `with`.

```
> with(linalg);
  M := matrix(2, 2, [[1, 2], [2, 5]]);
  M := 
$$\begin{bmatrix} 1 & 2 \\ 2 & 5 \end{bmatrix}$$

> inverse(M);

$$\begin{bmatrix} 5 & -2 \\ -2 & 1 \end{bmatrix}$$

> eigenvalues(M);
 $3+2\sqrt{2}, 3-2\sqrt{2}$ 
> LUdecomp(M, L = 'L', U = 'U');
  evalm(L);
  evalm(U);

$$\begin{bmatrix} 1 & 0 \\ 2 & 1 \end{bmatrix}$$


$$\begin{bmatrix} 1 & 2 \\ 0 & 1 \end{bmatrix}$$

> equal(M, multiply(L, U));
  true
```

Подробную информацию о пакете `linalg` можно получить, исполнив команду `?linalg`.

Лабораторная работа №3.Решение систем линейных уравнений.

Для решения матричных уравнений используется функция `linsolve`.

Решим систему линейных уравнений 
$$\begin{cases} x_1 + 2x_2 = 1, \\ 2x_1 + 5x_2 = 3. \end{cases}$$

Данную систему можно записать в виде 
$$\begin{pmatrix} 1 & 2 \\ 2 & 5 \end{pmatrix} \cdot x = \begin{pmatrix} 1 \\ 3 \end{pmatrix}.$$

```
> with(linalg):
  A := matrix(2, 2, [[1, 2], [2, 5]]):
  f := vector(2, [1, 3]):
  linsolve(A, f);
[-1,1]
```

Решите данную систему, не используя функцию `linsolve`.

Пусть теперь:  $A$  – матрица Гильберта;

$y$  – случайный вектор;

$f$  – вектор, равный произведению  $A$  на  $y$ .

Убедитесь, что решение  $x$  матричного уравнения  $Ax = f$  в точности совпадает с вектором  $y$ .

Преобразуйте компоненты матрицы  $A$  и вектора  $f$  к форме чисел с плавающей точкой, после чего ещё раз найдите решение уравнения  $Ax = f$ . Исследуйте погрешность полученного решения, изменяя размерность уравнения и значение системной переменной `Digits`.

**Константы**

Именованные константы Maple перечислены в **Приложении 5**. Подробную информацию о константах можно получить, исполнив команду `?constants`.

```
> constants;
false,g,∞,true,Catalan,FAIL,p
> type(false, constant);
true
> type(exp(I * Pi), constant);
true
```

**Переменные**

Для явного указания типа переменных используется конструкция `name::type`. Для присваивания переменной значения используется оператор `:=`. Для отмены присваивания – одна из конструкций `x := `x`` или `x := evaln(x)`.

Для придания переменным статуса предполагаемых используется функция `assume`.



```

> sqrt(x^2);
  M := matrix([[x, x^2], [exp(x), 1]]):
  map(ln, M);

$$\begin{matrix} \sqrt{x^2} \\ \begin{bmatrix} \ln(x) & \ln(x^2) \\ \ln(e^x) & 0 \end{bmatrix} \end{matrix}$$

> assume(x, positive):
  sqrt(x^2);
  M := matrix([[x, x^2], [exp(x), 1]]):
  map(ln, M);

$$\begin{matrix} x \sim \\ \begin{bmatrix} \ln(x \sim) & 2 \ln(x \sim) \\ x \sim & 0 \end{bmatrix} \end{matrix}$$

> about(x);
Originally x, renamed x~:
  is assumed to be: RealRange(Open(0), infinity)

```

Функцию `assume` можно использовать не только для переменных, но и для целых выражений. Некоторые параметры этой функции можно найти в **Приложении 1**.

## Операторы и операнды

В Maple имеется три типа операторов: бинарные (`binary`), с одним операндом (`unary`) и без операндов (`nullary`). Все операторы Maple перечислены в **Приложении 6** в порядке их выполнения. Подробную информацию об операторах можно получить, исполнив команду `?operators`.

Для задания функций используется функциональный оператор `->`:

```

> f := x -> exp(-I * x) * sin(x):
  f(Pi / 2);

$$-I$$

> g := (x, y) -> sqrt(x^2 + y^2):
  g(3, 4);

$$5$$

> simplify(g(sin(x), cos(x)));

$$1$$


```

Функция `define` служит для создания операторов. Функция `definemore` позволяет добавить свойства к уже определённой функцией `define` оператору.

```

> define(L, linear, L(1) = t):
  L(x^2 + 2 * x + 2);

$$L(x^2) + 2L(x) + 2t$$

> define(h, commutative):
  h(x, y) - h(y, x);

$$0$$


```

```
> define(f, diff(f(x), x) = 1 / f(x)):
  diff(f(exp(x)), x);

$$\frac{e^x}{f(e^x)}$$

> int(3 * f(x), x);

$$f(x)^3$$

```

Некоторые параметры функции `define` описаны в **Приложении 6**.

#### Лабораторная работа №4.

##### Создание операторов.

Создадим оператор, вычисляющий числа Фибоначчи:

```
> define(F, F(0) = 0, F(1) = 1,
  F(n::posint) = F(n - 2) + F(n - 1)):
  F(10);
  F(22);
55
17711
```

С помощью системной функции `time` определим время, затраченное на вычисление значения `F(22)`:

```
> time(F(22));
1.515
```

Это время оказалось довольно значительным, поскольку каждое новое число Фибоначчи вычисляется заново. Создайте оператор, вычисляющий числа Фибоначчи без использования рекурсии.

## 3. Средства программирования в Maple

### Условный оператор

Условный оператор в Maple имеет следующую конструкцию:

```
if <Условие>
  then <Последовательность действий>
  { elif <Условие>
  then <Последовательность действий> }
  [ else <Последовательность действий> ]
end if;
> x := 5:
  if x < 0
  then print(`Negative`)
  elif x > 0
  then print(`Positive`)
  else print(`Zero`)
  end if;
Positive
```

## Операторы цикла

Цикл в Maple может иметь одну из следующих конструкций:

```

for <Имя> from <Выражение>
  [ by <Выражение> ] [ to <Выражение> ]
  [ while <Условие> ]
do
  <Последовательность действий>
end do;
for <Имя> in <Набор значений>
  [ while <Условие> ]
do
  <Последовательность действий>
end do;
while <Условие>
do
  <Последовательность действий>
end do;
> for i in [1, 2, 5, -1, 7, 12]
  while i > 0
  do
    print(i)
  end do;

1
2
5

```

Рассмотрим пример задания единичной матрицы:

```

> n := 3:
A := array(1 .. n, 1 .. n):
for i from 1 to n
do
  for j from 1 to n
  do
    if i = j
    then A[i, j] := 1
    else A[i, j] := 0
    end if
  end do
end do:
evalm(A);

```

$$\begin{bmatrix} 1 & 0 & 0 \\ 0 & 1 & 0 \\ 0 & 0 & 1 \end{bmatrix}$$

## Процедуры

Общая форма задания процедуры в Maple:

```
<Имя процедуры> := proc(<Список параметров>)
  [ local <Список локальных переменных>; ]
  [ global <Список глобальных переменных>; ]
  [ options <Список расширяющих ключей>; ]
  [ description <Комментарий к процедуре>; ]
  <Тело процедуры>
end proc;
```

Процедура возвращает значение последнего выражения в её теле.

Реализуем процедуру вычисления модуля комплексного числа:

```
> modc := proc(z)
  sqrt(Re(z)^2 + Im(z)^2)
end proc:
modc(3 + 4 * I);
```

5

При подготовке процедур надо предусматривать их поведение при возможных ошибках. При выявлении ошибки предусматривается вывод соответствующего сообщения. Для этого используется функция `error`.

Реализуем процедуру вычисления квадратного корня из действительного числа с использованием функции `error`:

```
> sqroot := proc(x)
  if not (type(x, numeric) or type(x, realcons))
    or signum(x) = -1
  then error "invalid variable x: %1", x
  else x^(1 / 2)
  end if
end proc:
sqroot(-2);
Error, (in sqroot) invalid variable x: -2
```

Расширяющий ключ `remember` обеспечивает занесение результатов обращений к процедуре в таблицу памяти. Этот ключ особенно полезен при реализации рекурсивных процедур.

Рассмотрим пример из *Лабораторной работы №4*. Реализуем процедуру вычисления чисел Фибоначчи, используя ключ `remember`:

```
> F := proc(n::nonnegint)
  if n < 2
    then n
    else F(n - 2) + F(n - 1)
  end if
end proc:
time(F(100));
```

0.

Подробную информацию о процедурах и ключах расширения процедур можно получить, исполнив команды `?proc` и `?options` соответственно.

## Создание библиотеки процедур

### Лабораторная работа №5.

#### Создание библиотеки процедур.

Создадим библиотеку процедур, работающих с комплексными числами, записанными в показательной форме. Вначале зададим пустую таблицу под будущие процедуры:

```
> ComplexExpLib := table([]):
```

Любое комплексное число  $z$  может быть записано в виде  $z = \text{abs}(z) \cdot e^{i \text{argument}(z)} = r \cdot e^{ij}$ . Будем представлять комплексное число  $z$  в качестве списка, состоящего из двух элементов:  $r$  и  $j$ .

Реализуем процедуру перевода комплексных чисел из алгебраической формы в показательную:

```
> ComplexExpLib[AlgToExp] := proc(z::algebraic)
    simplify([abs(z), argument(z)])
end proc:
```

Реализуйте следующие процедуры:

перевод комплексных чисел из показательной формы в алгебраическую; проверка равенства двух комплексных чисел; умножение и деление двух комплексных чисел; возведение комплексного числа в  $n$  степень; вычисление  $k$ -го значения корня  $n$  степени из комплексного числа по формуле Муавра.

После исполнения команды `with(ComplexExpLib)`, можно проверить работу написанных процедур.

Сохраните библиотеку на диск с помощью команды `save`:

```
> save(ComplexExpLib, `c:/ComplexExpLib.m`):
```

Чтобы использовать созданную библиотеку процедур, нужно выполнить последовательность команд:

```
> read(`c:/ComplexExpLib.m`):
with(ComplexExpLib):
```

## Работа с файлами

Для записи данных в файл служит функция `writedata`:

```
writedata[APPEND](filename, data, format);
```

Данные могут задаваться списком, вектором или матрицей, формат данных может быть `integer`, `float` или `string`. Необязательный указатель `APPEND` используется, если данные дописываются в уже созданный файл.

```
> data := array([[1, 2, 3, 4], [5, 6, 7, 8]]):
writedata(`c:/data.txt`, data, integer):
```

Считывание данных из файла обеспечивает функция `readdata`:

```
readdata(filename, format, n);
```

Здесь  $n$  – число считываемых столбцов.

```
> data := readdata(`c:/data.txt`, integer, 2);
data := [[1,2],[5,6]]
```

## 4. Средства графики системы Maple

### Двумерная графика

Для построения двумерных графиков служит процедура `plot`.

При построении графика функции, заданной явно, процедура `plot` записывается следующим образом:

```
plot(y(x), x = x1 .. x2, y1 .. y2, options);
```

При построении графика функции, заданной параметрически, процедура `plot` записывается следующим образом:

```
plot([x(t), y(t), t = t1 .. t2],
     x1 .. x2, y1 .. y2, options);
```


Для построения сеточной функции процедура `plot` записывается следующим образом:

```
plot(A, x1 .. x2, y1 .. y2, options);
```


Здесь функция задана матрицей `A`; первый столбец матрицы – значения аргумента, второй столбец матрицы – соответствующие значения функции.

Некоторые значения параметра `options` перечислены в **Приложении 7**.


```
> plot(sin(x) / x, x = -10 .. 10, color = black);
```




```
> plot([cos(x), sin(x) / x], x = - 2 * Pi, 2 * Pi,
     color = black);
```



```
> plot([sin(3 * t), cos(5 * t), t = 0 .. 2 * Pi],
     color = black);
```



```
> A := [[n, sin(n / 4)] $n = -12 .. 12]:
  plot(A, color = black, style = point);
```



## Трёхмерная графика

Для построения трёхмерных графиков служит процедура `plot3d`.

При построении графика поверхности, заданной явно, процедура `plot3d` записывается следующим образом:


```
plot3d(z(x, y), x = x1 .. x2, y = y1 .. y2,
  options);
```

При построении графика поверхности, заданной параметрически, процедура `plot3d` записывается следующим образом:


```
plot3d([x(s, t), y(s, t), z(s, t)], s = s1 .. s2,
  t = t1 .. t2, options);
```

Некоторые значения параметра `options` перечислены в **Приложении 7**.

```
> plot3d(sin(x^2 + y^2), x = -Pi / 2 .. Pi / 2,
  y = -Pi / 2 .. Pi / 2, color = white);
```




```
> plot3d(1, s = 0 .. 2 * Pi, t = 0 .. Pi,
  coords = spherical, color = white);
```




## Расширенные средства графики

Часто удобно использовать функции из графического пакета `plots`. Некоторые функции из этого пакета представлены в **Приложении 7**. Подробную информацию о пакете `plots` можно получить, исполнив команду `?plots`.


```
> with(plots):
  R := t -> 100 / (100 + (t - Pi / 2)^8) *
    (2 - sin(7 * t) - cos(30 * t) / 2):
  polarplot(R(t), t = - Pi / 2 .. 3 / 2 * Pi,
    color = black, axes = NONE);
```



```
> A := [seq([seq(sin(n * m / (20 * Pi)),
  n = -10 .. 10)], m = -15 .. 15)]:
  listplot3d(A, color = white, axes = BOXED);
```



```
> f := (x, y) -> x / sqrt(x^2 + y^2 + 1):
  g := (x, y) -> - y / sqrt(x^2 + y^2 + 1):
  fieldplot([f, g], -1 .. 1, -1 .. 1,
    arrows = SLIM);
```





## 5. Работа с полиномами

### Оценка степени полинома и его коэффициентов

Функция `randpoly` возвращает случайный полином; функция `collect` возвращает полином, объединяя коэффициенты при степенях указанной переменной:

```
> p := randpoly([x, y], degree = 2);
p := collect(p, x);
p := -61 + 41x - 58y - 90x^2 + 53xy - y^2
p := -90x^2 + (53y + 41)x - 61 - y^2 - 58y
> q := (x - 2 * y)^2 + 3 * x - 4 * (y^2 + y) + 5;
q := collect(q, x);
q := x^2 + (3 - 4y)x - 4y + 5
```

Некоторые функции, используемые для оценки степени полинома и его коэффициентов, представлены в **Приложении 8**.

```
> degree(p, x);
2
> degree(q, [x, y]);
2
> coeffs(q, [x, y]);
-4y + 5, 3 - 4y, 1
```

### Вычисление корней полинома

Для вычисления корней полиномов, зависящих от одной переменной, служит функция `roots`. Корни возвращаются в виде  $[[r_1, k_1], \dots, [r_n, k_n]]$ , где  $r_i$  – корень полинома кратности  $k_i$ :

```
> p := 4 * x^5 + x^3 + x^2 - 3 * x + 1;
roots(p, x);
factor(p);
[[ [1/2, 2], [-1, 1] ]
(x + 1)(x^2 + 1)(-1 + 2x)^2
```

Функция `realroot` возвращает интервалы, в которых находятся действительные корни полинома:

```
> realroot(p, x);
[[0, 2], [-2, 0]]
```

### Лабораторная работа №6.

#### Исследование полинома на устойчивость.

Исследуйте полином  $L(p) = a_0 + a_1x + \dots + a_nx^n$  с вещественными коэффициентами на устойчивость, воспользовавшись критерием Рауса-Гурвица.

## 6. Решение уравнений и неравенств

### Решение уравнений

Для решения уравнений в аналитическом виде используется функция `solve`:

```
> eqn := sqrt(ln(x)) = 2:
  solve(eqn, x);

$$e^4$$

> f := x -> sqrt(ln(x)) - 2:
  solve(f(x), x);

$$e^4$$

> eqn := sin(x) = 1 / 2:
  solve(eqn, x);

$$\frac{\pi}{6}$$

```

Чтобы функция `solve` выдавала все решения уравнения, глобальной переменной `_EnvAllSolutions` присваивают значение `true`:

```
> _EnvAllSolutions := true:
  eqn := sin(x) = 1 / 2:
  solve(eqn, x);

$$\frac{1}{6}\pi + \frac{2}{3}\pi - 2\pi \sim + 2\pi - \frac{1}{6}\pi \sim$$

```

В решениях могут встречаться следующие обозначения: `_Z` – целое число, `_NN` – целое неотрицательное число, `_B` – 0 или 1.

При решении систем уравнений они и список переменных задаются как множества. Функция `assign` обеспечивает присваивание переменным значений, полученных в результате решения системы:

```
> eqns := {x * y = 2, x + y = 3}:
  sol := solve(eqns, {x, y}):
  [x, y];
  assign(sol):
  [x, y];
  unassign('x', 'y'):
  [x, y]
  [1, 2]
```

Функцию подстановки `subs` применяют, когда нужно проверить, является ли определённое значение решением уравнения:

```
> eqn := x^2 - 3 * x + 2 = 0:
  sol := solve(eqn, x):
  subs(x = sol[1], eqn);
  subs(x = -1, eqn);
0=0
6=0
```

Иногда результаты решения уравнений представляются в компактном виде с помощью функции `RootOf`. В этом случае все корни можно получить с помощью функции `allvalues`:

```
> eqns := {x * y = 2 * a, x + y = 2 * b}:
  solve(eqns, {x, y});
  allvalues(%);
{x = -RootOf(_Z^2 - 2_Z b + 2a) + 2b, y = RootOf(_Z^2 - 2_Z b + 2a)}
{x = b - sqrt(b^2 - 2a), y = b + sqrt(b^2 - 2a)}, {x = b + sqrt(b^2 - 2a), y = b - sqrt(b^2 - 2a)}
```

С помощью функции `solve` можно решать и функциональные уравнения:

```
> eqn := sqrt(f(x)) = x:
  f := solve(eqn, f):
  f(x);
x^2
```

## Решение неравенств

Для решения неравенств в аналитическом виде, как и для уравнений, используется функция `solve`:

```
> eqn := sqrt(ln(x)) < 2:
  solve(eqn, x);
RealRange(1, Open(e^4))
```

К сожалению, в **Maple 8** функция `solve` устойчива при работе только с одним неравенством, и при поиске решения только по одной переменной:

```
> eqns := {x^2 + y^2 <= 1, x >= 0, y >= 0}:
  sol := solve(eqns, {x, y});
sol :=
```

## Решение уравнений в численном виде

Для решения уравнений в численном виде используется функция `fsolve`:

```
> f := x -> x^4 - 4 * x^3 + 2 * x^2 - 4 * x + 1:
  fsolve(f(x), x);
0.2679491924, 3.732050808
```

Функция `fsolve` позволяет задавать интервал, на котором ищется решение, а также искать корни полинома в комплексной форме:

```
> f := x -> x^4 - 4 * x^3 + 2 * x^2 - 4 * x + 1:
  fsolve(f(x), x = 2 .. 4);
3.732050808
> fsolve(f(x), x, complex);
-1.1, 1.1, 0.2679491924, 3.732050808
```

Аналогично решаются системы уравнений:

```
> f := (x, y) -> sin(x + y) - exp(x) * y:
  g := (x, y) -> x^2 - y - 2:
  fsolve({f(x, y), g(x, y)}, {x = 0, y = 0});
{x = -1.552838698, y = -0.6687012050}
```

Лабораторная работа №7.Решение уравнений в численном виде.

Создадим процедуру вычисления корней уравнения  $f(x)=0$  методом половинного деления интервала:

```

> floatsolve := proc(ff, xint, epsilon)
  local x, a, b, c, f;
  x := lhs(xint);
  a := op(rhs(xint))[1];
  b := op(rhs(xint))[2];
  f := unapply(evalf(ff), x);
  if f(a) * f(b) > 0
    then error "Неверно введённый интервал"
  else if f(a) * f(b) = 0
    then
      if f(a) = 0
        then c := a
      else c := b
      end if
    else
      while (abs(b - a) > epsilon)
        and (f(c) <> 0)
      do
        c := (a + b) / 2;
        if f(a) * f(c) > 0
          then a := c
        else b := c
        end if
      end do
    end if;
    evalf(c);
  end if;
end proc;
> f := x -> x^4 - 4 * x^3 + 2 * x^2 - 4 * x + 1;
fsolve(f(x), x = 2 .. 4);
floatsolve(f(x), x = 2 .. 4, 1E-4);
3.732050808
3.731994629

```

Модифицируйте процедуру floatsolve так, чтобы она объединяла в себе метод половинного деления и метод секущих. Для этого добавьте процедуре ещё один параметр, определяющий метод поиска решения, переопределите команду присваивания точке c значения и измените условие выхода из цикла while.

Убедитесь, что погрешность обоих методов имеет порядок  $O(e)$ .

## Решение уравнений в целых числах

Для решения уравнений в целых числах используется функция `isolve`:

```
> eqn := x * y = 1:
  isolve(eqn);
{x = -1, y = -1}, {x = 1, y = 1}
> eqn := 2 * x + 3 * y = 5:
  isolve(eqn);
{x = 1 - 3_Z1, y = 1 + 2_Z1}
```

## Решение рекуррентных уравнений

Для решения рекуррентных уравнений используется функция `rsolve`:

```
> eqn := {f(n) = n * f(n - 1), f(0) = 1}:
  rsolve(eqn, f);
G(n+1)
> eqn := {F(n + 2) = F(n + 1) + F(n),
  F(0) = 0, F(1) = 1}:
  rsolve(eqn, F);

$$\frac{\sqrt{5}\left(\frac{1}{2} + \frac{\sqrt{5}}{2}\right)^n}{5} - \frac{\sqrt{5}\left(\frac{1}{2} - \frac{\sqrt{5}}{2}\right)^n}{5}$$

```

## 7. Интерполирование функций

### Полиномиальная интерполяция


Функция `interp` создаёт полином минимально необходимой степени, проходящий через заданные точки. При этом интерполируемая функция задаётся двумя векторами: вектором абсцисс и вектором ординат.

Построим интерполяционный полином для функции  $y(x) = \frac{\sin(x)}{x}$  на равномерной сетке на отрезке  $[0, 6\pi]$ :

```
> y := x -> if x = 0
  then 1
  else sin(x) / x
  end if:
T := 6 * Pi:
N := 7:
X := [(T * n / N) $n = 0 .. N]:
Y := map(y, X):
L := x -> interp(X, Y, x):
evalf(L(x), 3);
1. + 0.94x - 1.2x2 + 0.38x3 - 0.056x4 + 0.0042x5 - 0.00016x6 + 0.27 10-5 x7
```

Построим график получившегося интерполяционного полинома:


```
> A := [[T * n / N, y(T * n / N)] $n = 1 .. N]:
plot([y, L, A], 0 .. T, color = black,
style = [LINE, LINE, POINT]);
```



Как видно из графика, на границах области интерполирования отклонения полинома существенны. Интерполяционные свойства полинома можно улучшить, выбрав узлы интерполирования следующим образом:

$X_n = \frac{(b-a)c_n + (b+a)}{2}$ , где  $c_n = \cos\left(\frac{(1+2n)\pi}{2(N+1)}\right)$  – нуль полинома Чебышева.

```
> y := x -> if x = 0
              then 1
              else sin(x) / x
            end if:
T := 6 * Pi:
N := 7:
C := n -> cos((1 + 2 * n) * Pi / (2 * (N + 1))):
X := [(T * C(n) + T) / 2] $n = 0 .. N]:
Y := map(y, X):
L := x -> interp(X, Y, x):
A := [[(T * C(n) + T) / 2,
y((T * C(n) + T) / 2)] $n = 1 .. N]:
plot([y, L, A], 0 .. T, color = black,
style = [LINE, LINE, POINT], symbol = CIRCLE);
```



Лабораторная работа №7.Интерполяция методом Ньютона.

Интерполяционный полином Ньютона имеет вид

$$y(x_0) + \sum_{n=1}^N w_{n-1}(x) y(X_0, X_1, \dots, X_n), \text{ где } w_n(x) = (x - X_0)(x - X_1) \dots (x - X_n).$$

Здесь  $y(X_0, X_1, \dots, X_n)$  – разделённая разность порядка  $n$ .

На отрезке  $[0, 6\pi]$  постройте интерполяционный полином Ньютона для функции  $y(x) = \frac{\sin(x)}{x}$ . Убедитесь, что он совпадает с интерполяционным полиномом, построенным с помощью функции `interp`.


**Интерполяция сплайнами**

При интерполяции сплайнами на каждом интервале интерполируемая функция представляется в виде полинома, своего для каждого интервала. Кроме того, на сплайн накладывается условие гладкости.

В Maple для выполнения интерполяции сплайнами используется функция `spline`:

```
> y := x -> if x = 0
              then 1
              else sin(x) / x
              end if:

T := 6 * Pi:
N := 7:
X := [(T * n / N) $n = 0 .. N]:
Y := map(y, X):
S1 := x -> spline(X, Y, x, linear):
S2 := x -> spline(X, Y, x, quadratic):
S3 := x -> spline(X, Y, x, cubic):
A := [[T * n / N, y(T * n / N)] $n = 1 .. N]:
plot([S1, S2, S3, A], 0 .. T, color = black,
     style = [LINE, LINE, LINE, POINT],
     linestyle = [DASHDOT, DASH, SOLID],
     symbol = CIRCLE);
```



## Общая задача интерполирования

### Лабораторная работа №8.

#### Общая задача интерполирования.

Будем искать интерполяционную функцию в виде  $F_N(x) = \sum_{n=0}^N a_n j_n(x)$ , где  $j_0(x), j_1(x), \dots, j_N(x)$  – базовые интерполяционные функции, которые должны быть независимыми. Тогда из условия совпадения значений интерполяционной и интерполируемой функций в узловых точках получим систему из  $N+1$  уравнения относительно  $N+1$  неизвестного коэффициента  $a_n$ :  $F_N(X_n) = Y_n, n = 0, \dots, N$ .

```
> interpolate := proc(X, Y, f)
  local n, N, L, t, eqns, a, s;
  N := linalg[vectdim](X) - 1;
  s := 0;
  for n from 0 to N
    do
      s := s + a[n] * f(n, t)
    end do;
  L := unapply(s, t);
  eqns := {};
  for n from 1 to N + 1
    do
      eqns := eqns union {Y[n] = L(X[n])}
    end do;
  solve(eqns, {seq(a[n], n = 0 .. N)});
  assign(%);
  L
end proc;
```

Протестируйте процедуру `interpolate` на следующих системах базовых интерполяционных функций:

степенные функции  $j_n(x) = x^n$ ;

полиномы Чебышева  $j_n(x) = \cos(n \arccos(x))$ ;

степенные функции  $j_n(x) = \begin{cases} \cos(nx), & n = 0, 2, 4, \dots; \\ \sin(nx), & n = 1, 3, 5, \dots \end{cases}$

В качестве интерполируемой функции возьмите табулированную функцию:

$$\begin{cases} X = [-2, -1, 1, 2], \\ Y = [-5, 0, 4, 15]. \end{cases}$$

Сравните полученные результаты. Постройте графики получившихся интерполяционных функций.



## 8. Дифференцирование функций

### Вычисление пределов функций

Для вычисления пределов функций используется функция `limit`:

```
> y := x -> (tan(x) - x) / (x - sin(x)):
  Limit(y(x), x = 0) = limit(y(x), x = 0);
```

$$\lim_{x \rightarrow 0} \frac{\tan(x) - x}{x - \sin(x)} = 2$$

```
> y := x -> (sin(x) / x)^(1 / (x^2)):
  Limit(y(x), x = 0) = limit(y(x), x = 0);
```

$$\lim_{x \rightarrow 0} \left( \frac{\sin(x)}{x} \right)^{\left( \frac{1}{x^2} \right)} = e^{\left( \frac{-1}{6} \right)}$$

```
> y := x -> cot(x):
  Limit(y(x), x = 0, right) =
  limit(y(x), x = 0, right);
```

$$\lim_{x \rightarrow 0^+} \cot(x) = \infty$$

```
> Limit(y(x), x = 0, left) =
  limit(y(x), x = 0, left);
```

$$\lim_{x \rightarrow 0^-} \cot(x) = -\infty$$

### Исследование функций на непрерывность

Функция `iscont` позволяет исследовать функции на непрерывность:

```
> y := x -> x + 1 / x:
  iscont(y(x), x = 0 .. 1);
  true
> iscont(y(x), x = 0 .. 1, closed);
  false
```

Функция `discont` служит для поиска точек разрыва функций:

```
> discont(y(x), x);
  {0}
> u := x -> GAMMA(x):
  discont(u(x), x);
  { -_NN1 ~ }
```

Функция `singular` служит для поиска особых точек функций:

```
> singular(y(x), x);
  {x=0}, {x=∞}, {x=-∞}
> singular(u(x), x);
  {x=-_N1 ~ +1}, {x=∞}
```

## Вычисление производных функций

Для вычисления производных используется функция `diff`:

```
> y := x -> x + 1 / x:
```

```
Diff(y(x), x) = diff(y(x), x);
```

$$\frac{d}{dx} \left( 1 + \frac{1}{x} \right) = 1 - \frac{1}{x^2}$$

```
Diff(y(x), x$2) = diff(y(x), x$2);
```

$$\frac{d^2}{dx^2} \left( 1 + \frac{1}{x} \right) = \frac{2}{x^3}$$

```
> u := (x, y, z) -> sin(x) * sin(y) * sin(z):
```

```
Diff(u(x, y, z), x$2, y) =
```

```
diff(u(x, y, z), x$2, y);
```

$$\frac{\partial^3}{\partial y \partial x^2} (\sin(x) \sin(y) \sin(z)) = -\sin(x) \cos(y) \sin(z)$$

Функция `diff` позволяет, например, построить график производной, но не позволяет вычислить значение производной в конкретной точке. Этот недостаток устраняется созданием дифференциального оператора:

```
> D(y);
```

```
Diff(y, x) = D(y)(2);
```

$$x \rightarrow 1 - \frac{1}{x^2}$$

$$\left( \frac{\partial}{\partial x} y \right) (2) := \frac{3}{4}$$

Создадим оператор Лапласа  $D = \frac{\partial^2}{\partial x^2} + \frac{\partial^2}{\partial y^2} + \frac{\partial^2}{\partial z^2}$  для описанной выше функции  $u(x, y, z) = \sin(x) \sin(y) \sin(z)$ :

```
> Lu := (x, y, z) -> D[1$2](u)(x, y, z) +
```

```
D[2$2](u)(x, y, z) + D[3$2](u)(x, y, z):
```

```
Diff(u, x$2) + Diff(u, y$2) + Diff(u, z$2) =
```

```
Lu(x, y, z);
```

$$\left( \frac{\partial^2}{\partial x^2} u \right) + \left( \frac{\partial^2}{\partial y^2} u \right) + \left( \frac{\partial^2}{\partial z^2} u \right) = -3 \sin(x) \sin(y) \sin(z)$$

### Лабораторная работа №9.

#### Численное дифференцирование функций одной переменной с использованием интерполяционных формул.

Продифференцируйте интерполяционный полином Ньютона, построенный в *Лабораторной работе №7*. Найдите значения производных в узлах интерполирования. Исследуйте зависимость погрешности численного дифференцирования от величины  $h = \max_{0 \leq n \leq N-1} (X_n - X_{n-1})$ .

Лабораторная работа №10.Численное дифференцирование функций одной переменной.

Пусть  $X = [X_0, X_1, \dots, X_N]$  – вектор, задающий разбиение отрезка  $[a, b]$ :

$$a = X_0 < X_1 < \dots < X_N = b.$$

Пусть  $Y = [Y_0, Y_1, \dots, Y_N]$  – вектор соответствующих значений заданной функции  $y(x)$ :  $Y_n = y(X_n)$ ,  $n = 0, 1, \dots, N$ .

Для каждого  $n = 1, 2, \dots, N-1$  постройте

а) параболу,

б) окружность,

проходящую через точки  $(X_{n-1}, Y_{n-1})$ ,  $(X_n, Y_n)$  и  $(X_{n+1}, Y_{n+1})$ .

Вычислите значения производных построенных кривых в точках  $X_1, X_2, \dots, X_{N-1}$ . Сравните получившиеся значения с точными значениями производной функции  $y(x)$  в точках  $X_1, X_2, \dots, X_{N-1}$ .

Опишем процедуру, вычисляющую производную функции, заданной параметрически:

```
> parametricdiff := proc(x, y, t)
    diff(y, t) / diff(x, t)
end proc:
x := t -> t * cos(t):
y := t -> t * sin(t):
dy := t -> parametricdiff(x(t), y(t), t):
'dy / dx' = dy(t);

$$\frac{\partial}{\partial x} y = \frac{\sin(t) + t \cos(t)}{\cos(t) - t \sin(t)}$$

```

Как и в случае с функцией `diff`, невозможно вычислить значение таким образом описанной производной в конкретной точке. Поэтому и здесь правильнее будет создать дифференциальный оператор:


```
> parametricD := proc(x, y)
    D(y) / D(x)
end proc:
dy := parametricD(x, y)(t):
Diff(y, x) = dy(t);

$$\frac{\partial}{\partial x} y = \frac{\sin(t) + t \cos(t)}{\cos(t) - t \sin(t)}$$

```

Построим теперь график функции и её производной:

```
> T := 3 * Pi:
plot([[x, y, 0 .. T], [x, dy, 0 .. T]],
-T .. T, -T .. T, color = black,
linestyle = [DASHDOT, SOLID], axes = FRAME);
```



### Лабораторная работа №11.

#### Дифференцирование неявно заданных функций.

Производная неявно заданной функции вычисляется с помощью функции `implicitdiff`.

Найдём производную  $\frac{\partial y}{\partial x}$ , если функция задана уравнением  $x^2 + 4y^2 = 1$ :

```
> F := x^2 + 4 * y^2 = 1:
Diff(y, x) = implicitdiff(F, y, x);

$$\frac{\partial}{\partial x} y = -\frac{x}{4y}$$

```

Найдём производные  $\frac{\partial z}{\partial x}$  и  $\frac{\partial^2 z}{\partial x^2}$ , если функция задана системой уравнений

$$\begin{cases} x^2 + y^2 = z, \\ x^2 - xy + y^2 = 1: \end{cases}$$

```
> F1 := x^2 + y^2 = z:
F2 := x^2 - x * y + y^2 = 1:
Diff(z, x) =
simplify(implicitdiff({F1, F2}, {y, z}, z, x));
Diff(z, x$2) =
factor(implicitdiff({F1, F2}, {y, z}, z, x$2));

$$\frac{\partial}{\partial x} z = \frac{2(x^2 - y^2)}{x - 2y}$$


$$\frac{\partial^2}{\partial x^2} z = \frac{2(5x^3 - 12yx^2 + 15y^2x - 4y^3)}{(x - 2y)^3}$$

```

Используя теорему о неявной функции, постройте дифференциальные операторы для рассмотренных выше неявных функций.

## Замена переменных

Чтобы выполнить замену переменных в выражении, содержащем производные, используют функцию `dchange` из пакета `PDEtools`.

В уравнении  $x^2 \frac{d^2 y}{dx^2} + x \frac{dy}{dx} + y = 0$  выполним замену переменной  $x = e^t$ :

```
> with(PDEtools):
  eqn := x^2 * (D@@2)(y)(x) + x * D(y)(x) +
  y(x) = 0:
  simplify(dchange(x = exp(t), eqn));
```

$$\left( \frac{d^2}{dt^2} y(t) \right) + y(t) = 0$$

Выразим оператор Лапласа  $D = \frac{\partial^2}{\partial x^2} + \frac{\partial^2}{\partial y^2}$  через полярные координаты:

```
> Lu := (x, y) -> D[1$2](u)(x, y) +
  D[2$2](u)(x, y):
  tr := {x = r * cos(phi), y = r * sin(phi)}:
  simplify(dchange(tr, Lu(x, y), {r, phi}));
```

$$\frac{\left( \frac{d^2}{df^2} u(f, r) \right) + \left( \frac{d}{dr} u(f, r) \right) r + \left( \frac{d^2}{dr^2} u(f, r) \right) r^2}{r^2}$$

## Поиск экстремумов функций

Функция `extrema` служит для поиска локальных условных и безусловных экстремумов функций:

```
> y := x -> x + 1 / x:
  extrema(y(x), {}, x);
  {-2, 2}
> z := (x, y) -> (x - 3)^2 + (y + 4)^2 - 12:
  extrema(z(x, y), {x + y = 1}, {x, y});
  {-10}
```

К сожалению, функция `extrema` предназначена для решения задач на условный экстремум с ограничениями только типа равенств.

Для поиска глобальных безусловных минимумов и максимумов функций служат функции `minimize` и `maximize`:

```
> minimize(y(x), x = 0 .. infinity);
  2
> minimize(z(x, y), location);
  -12, {{{x = 3, y = -4}, -12}}
```

Лабораторная работа №12.Поиск безусловных экстремумов функций одной переменной.

Опишем процедуру поиска локальных безусловных экстремумов функций одной переменной:

```

> ExtrPoints := proc(f)
  local x;
  for x in {solve(D(f)(x) = 0, x)}
  do
    if evalf(D[1$2](f)(x)) = 0
    then print(cat("x = ", x, ": ?"))
    elif evalf(D[1$2](f)(x)) > 0
    then print(cat("x = ", x, ": Minimum"))
    else print(cat("x = ", x, ": Maximum"))
    end if
  end do
end proc:
> y := x -> 12 * x^5 + 15 * x^4 - 40 * x^3:
  ExtrPoints(y);
" x = -2 : Maximum"
" x = 0 : ?"
" x = 1 : Minimum"

```

Доработайте процедуру ExtrPoints так, чтобы в результате её выполнения не оставалось точек, требующих дополнительного исследования.

Лабораторная работа №13.Поиск безусловных экстремумов функций нескольких переменных.

Создайте процедуру поиска локальных безусловных экстремумов функций нескольких переменных. Достаточным условием минимума функции нескольких переменных является положительная определённость матрицы вторых производных этой функции, достаточное условие максимума – отрицательная определённость этой матрицы.

Лабораторная работа №14.Поиск условных экстремумов функций нескольких переменных.

## Приложение 1

### Функции, используемые при работе с выражениями

Таблица 1.1

Возможные значения параметра `form` функции преобразования выражений `convert(expr, form)`

Параметр	Описание
<code>float</code>	Преобразование в формат числа с плавающей точкой
<code>rational</code>	Преобразование в формат рационального числа
<code>fraction</code>	Преобразование в формат рациональной дроби
<code>decimal</code>	Преобразование числа в десятичный формат
<code>binary</code>	Преобразование числа в двоичный формат
<code>octal</code>	Преобразование числа в восьмеричный формат
<code>hex</code>	Преобразование числа в шестнадцатеричный формат
<code>arabic</code>	Переход от римских цифр к арабским
<code>roman</code>	Переход от арабских цифр к римским
<code>radians</code>	Переход от градусов к радианам
<code>degrees</code>	Переход от радиан к градусам
<code>vector</code>	Преобразование в векторный тип
<code>matrix</code>	Преобразование в матричный тип
<code>`+`</code>	Преобразование в сумму
<code>`*`</code>	Преобразование в произведение
<code>exp</code>	Переход от тригонометрических и гиперболических функций к экспоненциальным
<code>expln</code>	Переход от элементарных функций к экспоненциальным и тригонометрическим
<code>expsincos</code>	Переход от тригонометрических функций к синусу и косинусу, от гиперболических функций к экспоненциальным
<code>ln</code>	Переход от обратных тригонометрических и обратных гиперболических функций к логарифмическим
<code>sincos</code>	Переход от тригонометрических функций к синусу и косинусу, от гиперболических функций к гиперболическим синусу и косинусу
<code>tan</code>	Переход от тригонометрических и гиперболических функций к тангенсу
<code>trig</code>	Переход от экспоненциальных функций к тригонометрическим и гиперболическим
<code>GAMMA</code>	Переход от факториалов и биномиальных коэффициентов к гамма-функции
<code>factorial</code>	Переход от гамма-функции и биномиальных коэффициентов к факториалам
<code>binomial</code>	Переход от гамма-функции и факториалов к биномиальным коэффициентам
<code>signum</code>	Переход от модулей к функции знака
<code>abs</code>	Переход от функции знака и кусочно-гладких функций к модулям
<code>Heaviside</code>	Переход от кусочно-гладких функций, функции знака и модулей к функции Хевисайда
<code>piecewise</code>	Переход от функции Хевисайда, функции знака и модулей к кусочно-гладким функциям

Таблица 1.2

Возможные значения параметра `prop` функции `assume(expr, prop)`

Параметр	Описание
<code>numeric</code>	Число
<code>real</code>	Действительное число
<code>RealRange(x, y)</code>	Действительное число из отрезка $[a, b]$
<code>positive</code>	Положительное число
<code>negative</code>	Отрицательное число
<code>nonnegative</code>	Неотрицательное число
<code>nonpositive</code>	Неположительное число
<code>float</code>	Вещественное число с плавающей точкой
<code>rational</code>	Рациональное число
<code>irrational</code>	Иррациональное число
<code>fraction</code>	Рациональная дробь
<code>integer</code>	Целое число
<code>posint</code>	Положительное целое число
<code>negint</code>	Отрицательное целое число
<code>nonnegint</code>	Неотрицательное целое число
<code>nonposint</code>	Неположительное целое число
<code>prime</code>	Простое число
<code>even</code>	Чётное число
<code>odd</code>	Нечётное число
<code>complex</code>	Комплексное число
<code>nonreal</code>	Комплексное число, не лежащее на действительной оси
<code>imaginary</code>	Мнимое число
<code>GaussianInteger</code>	Комплексное число с целыми действительной и мнимой частями
<code>constant</code>	Константа
<code>realcons</code>	Действительная константа
<code>complexcons</code>	Комплексная константа
<code>scalar</code>	Скалярная величина
<code>vector</code>	Вектор
<code>matrix</code>	Матрица
<code>SquareMatrix</code>	Квадратная матрица
<code>triangular</code>	Треугольная матрица
<code>LowerTriangular</code>	Нижнетреугольная матрица
<code>UpperTriangular</code>	Верхнетреугольная матрица
<code>tridiagonal</code>	Трёхдиагональная матрица
<code>diagonal</code>	Диагональная матрица
<code>symmetric</code>	Симметрическая матрица, $A = A^T$
<code>antisymmetric</code>	Кососимметрическая матрица, $A = -A^T$
<code>idempotent</code>	Идемпотентная матрица, $A = A^2$
<code>singular</code>	Вырожденная матрица, $\det(A) = 0$
<code>PositiveDefinite</code>	Положительно определённая матрица, $x^T Ax > 0 \quad \forall x \neq q$



<code>mapping</code>	Функция
<code>unary</code>	Функция одного параметра
<code>binary</code>	Функция двух параметров
<code>EvenMap</code>	Чётная функция
<code>OddMap</code>	Нечётная функция
<code>monotonic</code>	Монотонная функция
<code>StrictlyMonotonic</code>	Строго монотонная функция
<code>continuous</code>	Непрерывная функция
<code>differentiable</code>	Дифференцируемая функция
<code>Infinitely-Differentiable</code>	Бесконечно дифференцируемая функция
<code>PolynomialMap</code>	Полином
<code>LinearMap</code>	Линейная функция

Таблица 1.3

Функции, используемые для оценивания выражений различного типа

Функция	Описание
<code>eval(expr, eqns)</code>	Символьное вычисление значения <code>expr</code> при заданных условиях <code>eqns</code>
<code>evalf(expr)</code>	Численное вычисление значения <code>expr</code> , вывод вычисленного значения в форме числа с плавающей точкой с мантиссой, порядок которой хранится в системной переменной <code>Digits</code>
<code>evalf(expr, n)</code>	Численное вычисление значения <code>expr</code> , вывод вычисленного значения в форме числа с плавающей точкой с мантиссой порядка <code>n</code>
<code>evalhf(expr)</code>	Численное вычисление значения <code>expr</code> , вывод вычисленного значения с точностью, присущей данной компьютерной системе
<code>evalm(mexpr)</code>	Вычисление значения матричного выражения <code>mexpr</code>
<code>evalc(cexpr)</code>	Вычисление значения комплексного выражения <code>cexpr</code>
<code>evalb(bexpr)</code>	Вычисление значения логического выражения <code>bexpr</code>
<code>evalr(rexpr)</code>	Оценка области значений интервального выражения <code>rexpr</code>
<code>shake(expr, ampl)</code>	Преобразование выражения <code>expr</code> в интервальное выражение (все константы <code>const</code> в <code>expr</code> заменяются на <code>const * INTERVAL(1 - Float(10, -ampl) .. 1 + Float(10, -ampl))</code> ); оценка области значений полученного интервального выражения

## Приложение 2

### Типы данных

Таблица 2.1  
Типы данных

Тип	Описание
numeric	Число
positive	Положительное число
negative	Отрицательное число
nonnegative	Неотрицательное число
nonpositive	Неположительное число
float	Вещественное число с плавающей точкой
rational	Рациональное число
fraction	Рациональная дробь
integer	Целое число
posint	Положительное целое число
negint	Отрицательное целое число
nonnegint	Неотрицательное целое число
nonposint	Неположительное целое число
prime	Простое число
even	Чётное число
odd	Нечётное число
complex	Комплексное число
nonreal	Комплексное число, не лежащее на действительной оси
imaginary	Мнимое число
constant	Константа
realcons	Действительная константа
complexcons	Комплексная константа
scalar	Скалярная величина
set	Множество
list	Список
vector	Вектор
matrix	Матрица
boolean	Выражение логического типа
function	Функция
symmfunc	Симметрическая функция
evenfunc	Чётная функция
oddfunc	Нечётная функция
trig	Тригонометрическая функция
arctrig	Обратная тригонометрическая функция
trigh	Гиперболическая функция
arctrigh	Обратная гиперболическая функция
polynom	Полином
linear	Линейная функция
quadratic	Квадратичная функция
cubic	Кубическая функция
quartic	Функция четвёртого порядка

## Приложение 3

### Математические функции

Таблица 3.1

#### Основные математические функции

Функция	Описание	Функция	Описание
$\sin(x)$	Синус	$\sinh(x)$	Синус гиперболический
$\cos(x)$	Косинус	$\cosh(x)$	Косинус гиперболический
$\tan(x)$	Тангенс	$\tanh(x)$	Тангенс гиперболический
$\cot(x)$	Котангенс	$\coth(x)$	Котангенс гиперболический
$\sec(x)$	Секанс	$\operatorname{sech}(x)$	Секанс гиперболический
$\csc(x)$	Косеканс	$\operatorname{csch}(x)$	Косеканс гиперболический
$\arcsin(x)$	Арксинус	$\operatorname{arsinh}(x)$	Арксинус гиперболический
$\arccos(x)$	Арккосинус	$\operatorname{arcosh}(x)$	Арккосинус гиперболический
$\arctan(x)$	Арктангенс	$\operatorname{artanh}(x)$	Арктангенс гиперболический
$\operatorname{arccot}(x)$	Арккотангенс	$\operatorname{arcoth}(x)$	Арккотангенс гиперболический
$\operatorname{arcsec}(x)$	Арксеканс	$\operatorname{arcsech}(x)$	Арксеканс гиперболический
$\operatorname{arccsc}(x)$	Арккосеканс	$\operatorname{arccsch}(x)$	Арккосеканс гиперболический

Функция	Описание
$\exp(x)$	Экспоненциальная функция
$\ln(x)$	Логарифм натуральный
$\log[b](x)$	Логарифм по основанию $b$
$\log_{10}(x)$	Логарифм десятичный
$\sqrt{x}$	Корень квадратный
$\operatorname{abs}(x)$	Модуль
$\operatorname{signum}(x)$	Знак
$\operatorname{round}(x)$	Округлённое значение
$\operatorname{ceil}(x)$	Наименьшее целое, большее или равное $x$
$\operatorname{floor}(x)$	Наибольшее целое, меньшее или равное $x$
$\operatorname{trunc}(x)$	Целая часть
$\operatorname{frac}(x)$	Дробная часть
$\max(x_1, x_2, \dots, x_N)$	Максимальное из чисел $x_1, x_2, \dots, x_N$
$\min(x_1, x_2, \dots, x_N)$	Минимальное из чисел $x_1, x_2, \dots, x_N$

Таблица 3.2

#### Основные функции для работы с целыми числами

Функция	Описание
$\operatorname{factorial}(n)$	Факториал
$\operatorname{iquo}(n, m)$	Целочисленное частное при делении $n$ на $m$
$\operatorname{irem}(n, m)$	Остаток от деления $n$ на $m$
$\operatorname{igcd}(n_1, n_2, \dots, n_N)$	Наибольший общий делитель
$\operatorname{lcm}(n_1, n_2, \dots, n_N)$	Наименьшее общее кратное
$\operatorname{ifactor}(n)$	Разложение на простые множители
$\operatorname{isprime}(n)$	Проверка, является ли число $n$ простым

Таблица 3.3

## Основные функции для работы с комплексными числами

Функция	Описание
$\text{Re}(z)$	Действительная часть
$\text{Im}(z)$	Мнимая часть
$\text{abs}(z)$	Модуль
$\text{argument}(z)$	Главное значение аргумента
$\text{conjugate}(z)$	Комплексно-сопряжённое число

Таблица 3.4

## Некоторые специальные математические функции

Функция	Описание
$\text{Dirac}(x)$	Дельта-функция Дирака $d(x) = \begin{cases} d(0), & x = 0; \\ 0, & x \neq 0 \end{cases}$
$\text{Heaviside}(x)$	Функция Хевисайда $h(x) = \begin{cases} 1, & x > 0; \\ 0, & x < 0 \end{cases}$
$\text{GAMMA}(x)$	Гамма-функция Эйлера $G(x) = \int_0^{\infty} e^{-t} t^{x-1} dt$
$\text{Beta}(x, y)$	Бета-функция Эйлера $B(x, y) = \int_0^1 t^{x-1} (1-t)^{y-1} dt$
$\text{Si}(x)$	Интегральный синус $\text{Si}(x) = \int_0^x \frac{\sin(t)}{t} dt$
$\text{Ci}(x)$	Интегральный косинус $\text{Ci}(x) = g + \ln(x) + \int_0^x \frac{\cos(t) - 1}{t} dt$
$\text{Shi}(x)$	Интегральный гиперболический синус $\text{Shi}(x) = \int_0^x \frac{\sinh(t)}{t} dt$
$\text{Chi}(x)$	Интегральный гиперболический косинус $\text{Chi}(x) = g + \ln(x) + \int_0^x \frac{\cosh(t) - 1}{t} dt$
$\text{Ei}(n, x)$	Интегральная экспоненциальная функция $\text{Ei}(n, x) = \int_1^{\infty} \frac{e^{-xt}}{t^n} dt, n \geq 0, x > 0$
$\text{Ei}(x)$	Интегральная экспоненциальная функция $\text{Ei}(x) = \text{vp} \int_{-\infty}^x \frac{e^t}{t} dt = g + \ln(x) + \int_0^x \frac{e^t - 1}{t} dt, x > 0$
$\text{Li}(x)$	Интегральный логарифм $\text{Li}(x) = \text{vp} \int_0^x \frac{dt}{\ln(t)} = \text{Ei}(\ln(x)), x > 1$

## Приложение 4

### Процедуры и функции, используемые для работы с векторами и матрицами

Таблица 4.1

Функции из пакета `linalg`, используемые для работы с векторами

Функция	Описание
<code>vector(n, list)</code>	Создание вектора размерности $n$ из списка <code>list</code>
<code>randvector(n)</code>	Создание случайного вектора размерности $n$
<code>vectdim(V)</code>	Размерность
<code>equal(U, V)</code>	Проверка совпадения векторов $U$ и $V$
<code>norm(V, k)</code>	Норма $\ V\ _k = \sqrt[k]{\sum_{j=1}^n  v_j ^k}$ , $k \geq 1$ , $vect\ dim(V) = n$
<code>norm(V, infinity)</code>	Норма $\ V\ _\infty = \max_{1 \leq j \leq n}  v_j $ , $vect\ dim(V) = n$
<code>normalize(V)</code>	Нормированный вектор $normalize(V) = \frac{V}{\ V\ _2}$
<code>dotprod(U, V)</code>	Скалярное произведение, $vect\ dim(U) = vect\ dim(V)$
<code>crossprod(U, V)</code>	Векторное произведение, $vect\ dim(U) = vect\ dim(V) = 3$
<code>angle(U, V)</code>	Угол между векторами $U$ и $V$ , $vect\ dim(U) = vect\ dim(V)$

Таблица 4.2

Функции из пакета `linalg`, используемые для работы с матрицами

Функция	Описание
<code>matrix(m, n, list)</code>	Создание матрицы размерности $m$ на $n$ из списка <code>list</code>
<code>randmatrix(m, n)</code>	Создание случайной матрицы размерности $m$ на $n$
<code>hilbert(n)</code>	Создание матрицы Гильберта размерности $n$ на $n$
<code>rowdim(A)</code>	Число строк
<code>coldim(A)</code>	Число столбцов
<code>row(A, i)</code>	Строка с номером $i$
<code>col(A, j)</code>	Столбец с номером $j$
<code>minor(A, i, j)</code>	Минор, полученный вычеркиванием строки с номером $i$ и столбца с номером $j$
<code>augment(A1, A2, ... , AN)</code>	Горизонтальное слияние матриц, $row\ dim(A_1) = row\ dim(A_2) = \dots = row\ dim(A_N)$
<code>stack(A1, A2, ... , AN)</code>	Вертикальное слияние матриц, $col\ dim(A_1) = col\ dim(A_2) = \dots = col\ dim(A_N)$
<code>equal(A, B)</code>	Проверка совпадения матриц $A$ и $B$
<code>rank(A)</code>	Ранг матрицы $A$
<code>trace(A)</code>	След квадратной матрицы $A$ , $trace(A) = \sum_{i=1}^n a_{ii}$ , $row\ dim(A) = col\ dim(A) = n$
<code>det(A)</code>	Определитель квадратной матрицы $A$

<code>norm(A, 1)</code>	Норма $\ A\ _1 = \max_{1 \leq j \leq n} \sum_{i=1}^m  a_{ij} $ , $\begin{cases} \text{row dim}(A) = m, \\ \text{col dim}(A) = n \end{cases}$
<code>norm(A, infinity)</code>	Норма $\ A\ _\infty = \max_{1 \leq i \leq m} \sum_{j=1}^n  a_{ij} $ , $\begin{cases} \text{row dim}(A) = m, \\ \text{col dim}(A) = n \end{cases}$
<code>norm(A, frobenius)</code>	Норма $\ A\ _{\text{frobenius}} = \sqrt{\sum_{i=1}^m \sum_{j=1}^n  a_{ij} ^2}$ , $\begin{cases} \text{row dim}(A) = m, \\ \text{col dim}(A) = n \end{cases}$
<code>cond(A, normname)</code>	Число обусловленности квадратной матрицы A $\text{cond}(A, \text{norm name}) = \ A\ _{\text{norm name}} \cdot \ A^{-1}\ _{\text{norm name}}$
<code>transpose(A)</code>	Транспонированная матрица
<code>multiply(A1, A2, ... , AN)</code>	Умножение матриц, $\begin{cases} \text{row dim}(A_k) = \text{col dim}(A_{k+1}), & k = 1, \dots, N-1, \\ \text{col dim}(A_k) = \text{row dim}(A_{k+1}), & k = 1, \dots, N-1 \end{cases}$
<code>multiply(A, V)</code>	Умножение матрицы A на вектор V, $\text{col dim}(A) = \text{vect dim}(V)$
<code>inverse(A)</code>	Обратная матрица
<code>adjoint(A)</code>	Сопряжённая матрица, $A \cdot \text{adjoint}(A) = \det(A) \cdot E$
<code>LUdecomp(A, L = 'L', U = 'U')</code>	LU-разложение
<code>eigenvalues(A)</code>	Собственные значения
<code>jordan(A)</code>	Жорданова форма квадратной матрицы A

Таблица 4.3

Функции из пакета `linalg`, используемые для работы с векторными полями

Функция	Описание
<code>grad(expr, var)</code>	Градиент выражения <code>expr</code> по вектору переменных <code>var</code> $\nabla \text{expr} = \left[ \frac{\partial \text{expr}}{\partial \text{var}_1}, \frac{\partial \text{expr}}{\partial \text{var}_2}, \dots, \frac{\partial \text{expr}}{\partial \text{var}_n} \right], \text{vect dim}(\text{var}) = n$
<code>potential(f, var, 'F')</code>	Проверка существования потенциала векторного поля <code>f</code> . Если потенциал существует, то он будет задан в <code>F</code> , при этом $\nabla F = f$
<code>diverge(f, var)</code>	Дивергенция векторного поля <code>f</code> $\text{div}(f) = \text{dotprod}(\nabla, f), \nabla = \left[ \frac{\partial}{\partial \text{var}_1}, \frac{\partial}{\partial \text{var}_2}, \dots, \frac{\partial}{\partial \text{var}_n} \right],$ $\text{vect dim}(\text{var}) = \text{vect dim}(f) = n$
<code>curl(f, var)</code>	Ротор векторного поля <code>f</code> $\text{curl}(f) = \text{crossprod}(\nabla, f), \nabla = \left[ \frac{\partial}{\partial \text{var}_1}, \frac{\partial}{\partial \text{var}_2}, \frac{\partial}{\partial \text{var}_3} \right],$ $\text{vect dim}(\text{var}) = \text{vect dim}(f) = 3$

## Приложение 5

### Константы

Таблица 5.1

Именованные константы

Константа	Описание
Pi	Число $p = 3.141592654$
gamma	Константа Эйлера $g = \lim_{n \rightarrow \infty} \left( \sum_{i=1}^n \frac{1}{i} - \ln(n) \right) = 0.5772156649$
Catalan	Константа Каталана $\sum_{i=0}^{\infty} \frac{(-1)^i}{(2i+1)^2} = 0.9159655942$
infinity	Бесконечность $\infty$
true	Логическая константа «Истина»
false	Логическая константа «Ложь»
FAIL	Логическая константа «Неопределённость»

## Приложение 6

### Операторы и операнды

Таблица 6.1

Операторы для работы с данными множественного типа

Оператор	Описание
$S \cup T$	Объединение двух множеств $S$ и $T$
$S \cap T$	Пересечение двух множеств $S$ и $T$
$S - T$	Разность двух множеств $S$ и $T$
$S \subset T$	Проверка, является ли множество $S$ подмножеством множества $T$
$s \in S$	Проверка, является ли $s$ элементом множества $S$

Таблица 6.2

Операторы в порядке их выполнения

Оператор	Тип
<code>  </code>	binary
<code>:-</code>	binary
<code>::</code>	binary
<code>%</code>	nullary
<code>&amp;-operators</code>	unary(prefix) / binary
<code>.</code>	unary(prefix / postfix) / binary
<code>!</code>	unary(postfix)
<code>^ @</code>	binary
<code>* &amp;* / @ intersect</code>	binary
<code>+ - union minus</code>	binary
<code>mod</code>	binary
<code>subset</code>	binary
<code>..</code>	binary
<code>&lt; &lt;= &gt; &gt;= = &lt;&gt; in</code>	binary
<code>\$</code>	unary(prefix) / binary
<code>not</code>	unary(prefix)
<code>and</code>	binary
<code>or</code>	binary
<code>xor</code>	binary
<code>implies</code>	binary
<code>-&gt;</code>	binary
<code>,</code>	binary
<code>assuming</code>	binary
<code>:=</code>	binary

Таблица 6.3

Возможные значения параметра `prop` функции `define(oper, prop)`

Параметр	Описание
<code>linear</code>	Оператор, линейный по первому аргументу
<code>multilinear</code>	Оператор, линейный по всем аргументам
<code>commutative</code>	Коммутативный оператор, $f(x, y) = f(y, x)$
<code>associative</code>	Ассоциативный оператор, $f(x, f(y, z)) = f(f(x, y), z) = f(x, y, z)$
<code>identity</code>	Единичный оператор
<code>zero</code>	Нулевой оператор



## Приложение 7

### Средства графики

Таблица 7.1

Возможные значения параметра `options` процедуры `plot(..., options)`

Параметр	Возможные значения	Значение по умолчанию	Описание
<code>coords</code>	<code>cartesian</code> <code>polar</code> ...	<code>cartesian</code>	Система координат
<code>scaling</code>	<code>UNCONSTRAINED</code> <code>CONSTRAINED</code>	<code>UNCONSTRAINED</code>	Масштаб
<code>discont</code>	<code>false</code> <code>true</code>	<code>false</code>	Разбиение графика функции на промежутки непрерывности
<code>title</code>	<Строка>		Заголовок
<code>legend</code>	<Строка>		Легенда
<code>color</code>	<code>white</code> <code>black</code> ...	<code>red</code>	Цвет линий графика функции
<code>style</code>	<code>LINE</code> <code>POINT</code> <code>PATCH</code> <code>PATCHNOGRID</code>	<code>PATCH</code>	Способ отображения графика функции
<code>linestyle</code>	<code>SOLID</code> <code>DOT</code> <code>DASH</code> <code>DASHDOT</code>	<code>SOLID</code>	Тип линий
<code>thickness</code>	<Целое число от 0 до 15>	0	Толщина линий
<code>symbol</code>	<code>POINT</code> <code>CROSS</code> <code>CIRCLE</code> <code>BOX</code> <code>DIAMOND</code>	<code>DIAMOND</code>	Тип символов для отображения точек
<code>symbolsize</code>	<Целое положительное число>	10	Размер символов для отображения точек
<code>axes</code>	<code>NORMAL</code> <code>FRAME</code> <code>BOXED</code> <code>NONE</code>	<code>NORMAL</code>	Тип координатных осей
<code>labels</code>	<Список из двух строк>		Надписи для координатных осей
<code>xtickmarks</code>	<Целое положительное число>		Минимальное число отображаемых меток на оси абсцисс
<code>ytickmarks</code>	<Целое положительное число>		Минимальное число отображаемых меток на оси ординат

Таблица 7.2

Возможные значения параметра `options` процедуры `plot3d(..., options)`

Параметр	Возможные значения	Значение по умолчанию	Описание
<code>coords</code>	<code>rectangular</code> <code>cylindrical</code> <code>spherical</code> ...	<code>rectangular</code>	Система координат
<code>scaling</code>	<code>UNCONSTRAINED</code> <code>CONSTRAINED</code>	<code>UNCONSTRAINED</code>	Масштаб
<code>title</code>	<Строка>		Заголовок
<code>color</code>	<code>white</code> <code>black</code> ...		Цвет поверхности
<code>shading</code>	<code>XYZ</code> <code>XY</code> <code>Z</code> <code>ZGRAYSCALE</code> <code>ZHUE</code> <code>NONE</code>	<code>XYZ</code>	Способ раскраски поверхности
<code>style</code>	<code>LINE</code> <code>POINT</code> <code>HIDDEN</code> <code>PATCH</code> <code>WIREFRAME</code> <code>CONTOUR</code> <code>PATCHNOGRID</code> <code>PATCHCONTOUR</code>	<code>PATCH</code>	Способ отображения поверхности
<code>linestyle</code>	<code>SOLID</code> <code>DOT</code> <code>DASH</code> <code>DASHDOT</code>	<code>SOLID</code>	Тип линий
<code>thickness</code>	<Целое число от 0 до 15>	0	Толщина линий
<code>symbol</code>	<code>POINT</code> <code>CROSS</code> <code>CIRCLE</code> <code>BOX</code> <code>DIAMOND</code>	<code>POINT</code>	Тип символов для отображения точек
<code>symbolsize</code>	<Целое положительное число>	10	Размер символов для отображения точек
<code>axes</code>	<code>NORMAL</code> <code>FRAME</code> <code>BOXED</code> <code>NONE</code>	<code>NONE</code>	Тип координатных осей
<code>labels</code>	<Список из трёх строк>		Надписи для координатных осей
<code>tickmarks</code>	<Список из трёх целых положительных чисел>		Минимальное число отображаемых меток на осях координат

Таблица 7.3

Функции из пакета `plots`, используемые для построения двумерных графиков

Функция	Описание
<code>display</code>	Построение графика для списка графических объектов
<code>animate</code>	Создание анимации графика функции
<code>polarplot</code>	Построение графика функции в полярной системе координат
<code>implicitplot</code>	Построение графика неявной функции
<code>complexplot</code>	Построение графика комплекснозначной функции
<code>listplot</code>	Построение графика сеточной функции
<code>pointplot</code>	Построение точечного графика
<code>polygonplot</code>	Построение многоугольника
<code>fieldplot</code>	Построение векторного поля
<code>gradplot</code>	Построение векторного поля градиента

Таблица 7.4

Функции из пакета `plots`, используемые для построения трёхмерных графиков

Функция	Описание
<code>display3d</code>	Построение графика для списка графических объектов
<code>animate3d</code>	Создание анимации графика функции
<code>spacecurve</code>	Построение пространственной кривой
<code>cylinderplot</code>	Построение графика функции в цилиндрической системе координат
<code>sphereplot</code>	Построение графика функции в сферической системе координат
<code>implicitplot3d</code>	Построение графика неявной функции
<code>complexplot3d</code>	Построение графика комплекснозначной функции
<code>listplot3d</code>	Построение графика сеточной функции
<code>pointplot3d</code>	Построение точечного графика
<code>polygonplot3d</code>	Построение многоугольника
<code>polyhedraplot</code>	Построение многогранника
<code>fieldplot3d</code>	Построение векторного поля
<code>gradplot3d</code>	Построение векторного поля градиента

Приложение 8**Функции, используемые при работе с полиномами**

Таблица 8.1

Функции, используемые для оценки степени и коэффициентов полинома

Функция	Описание
<code>degree(p, x)</code>	Степень полинома $p$ по переменной $x$
<code>ldegree(p, x)</code>	Наименьшая степень полинома $p$ по переменной $x$
<code>coeffs(p, x)</code>	Коэффициенты полинома $p$ при различных степенях $x$
<code>coeff(p, x, n)</code>	Коэффициент полинома $p$ при степени $x^n$
<code>lcoeff(p, x)</code>	Старший коэффициент полинома $p$ по переменной $x$
<code>tcoeff(p, x)</code>	Младший коэффициент полинома $p$ по переменной $x$